

UNITED STATES DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

HP-9845 Program for Computing and Plotting Full Schlumberger  
Sounding Curves and Master Sets Over Three Vertical Layers:  
Array Expanded at Right Angles to Strike.

by

Adel A.R. Zohdy

Open-File Report 80-248

1980

Citation of particular manufacturers and model numbers does not  
constitute endorsement by the U.S. Geological Survey.

Although this program has been tested extensively  
the U.S. Geological Survey cannot guarantee that it will give  
correct results in any or all particular applications.

HP-9845 Program for Computing and Plotting Full Schlumberger  
Sounding Curves and Master Sets Over Three Vertical Layers:  
Array Expanded at Right Angles to Strike.

By Adel A.R. Zohdy

INTRODUCTION

The three vertical-layers (or vertical-dike) problem was solved several decades ago by the tedious method of images (Hedstrom, 1932; Van Nostrand and Cook, 1966). Logn (1954) derived formulas (in the form of integral equations) for the potential and apparent resistivity of a pole-dipole or half-Schlumberger array. His approach for solving the boundary value problem for vertical layers is similar to that used by Stefanescu and others (1930) for solving the horizontal layer problem. Unfortunately, Logn was unable to evaluate the integral expression he derived without resorting to an infinitesimally thin-dike approximation. Kunetz (1955) also used the infinitesimally thin-dike model for evaluating his integral equation. Thus until now, the infinite-series expressions (image method) have been used for solving the thick-dike problem, and the integral equation evaluation has been used for solving the infinitesimally thin-dike problem.

The computer program presented in this report (see appendix II) is based on a novel method that uses the Gauss-Laguerre numerical integration formula. The integrals are easily evaluated for a dike with any desired width. The necessary equations for computing the Schlumberger apparent resistivity, using a symmetric (AMNB) and idealized ( $MN \rightarrow 0$ ) array, are given without derivation in appendix I. The equations are in a form suitable for the direct use of the Gauss-Laguerre numerical integration formula.

## BRIEF PROGRAM DESCRIPTION

This program is named: "VCSCH1" (Vertical Contact Schlumberger, version 1), and it is written in HP Enhanced BASIC.

The program can be used to compute and plot either a single sounding curve or to generate sets of master curves. Prompting statements, diagrams, illustrations, and appropriate error messages are displayed on the screen to assist the user at all steps.

The graphical presentation of any single sounding curve or set of master curves is made on a log-log plot of 4x3 cycles. The length of each log cycle is approximately 4 cm. The axes of each plot are appropriately labeled, and a label designating the distances ( $d_1$  and  $d_2$ ) and resistivities ( $\rho_1$ ,  $\rho_2$ , and  $\rho_3$ ) are given at the top of each plot, and a sketch of the dike and location of the sounding center is drawn in the upper right hand corner of the plot. The computation of any sounding curve is made at the rate 20 points per log cycle. Also the computations always include the points at which the electrode spacing is equal to the distance to a contact.

To compute a single sounding curve the user must type the numerical values (separated by commas) for: (i) the medium on which the sounding center is located (1, 2, or 3), (ii) the resistivity values for  $\rho_1$ ,  $\rho_2$ , and  $\rho_3$ , (iii) the distances to the contacts  $d_1$  and  $d_2$ , and (iv) the smallest and largest electrode spacing ( $AB/2$ ) values. The user must type each of these data values after the proper prompting statement appears on the screen and he must press CONT (continue) after each set of values is typed.

To compute a set of master curves, the user must type only the numerical values for: (i) the medium on which the sounding center is located, and (ii) the resistivities ( $\rho_1$ ,  $\rho_2$ , and  $\rho_3$ ). The resistivity value for the medium where the center is located is taken as unity. The values for  $d_1$  and  $d_2$  ( $d_1=1$ ,  $d_2=1, 1.5, 2, 4, 8, 20, 40$ , and infinity) are assigned by the program and are not entered by the user. Similarly the values for the smallest and largest electrode spacings ( $AB/2$ ) are assigned the values of 0.1 and 1000, and are not entered by the user. Note that for  $d_1=1$  and  $d_2=1, 1.5, 2, 4, 8, 20, 40$ , and infinity, and with the center on medium 1, the corresponding dike widths are : 0, 0.5, 1, 3, 7, 19, 39, and infinity, respectively; whereas with the center on medium 2 the corresponding dike widths are: 2, 2.5, 3, 5, 9, 21, 41, and infinity, respectively.

## INSTRUCTIONS FOR USERS

(1) Turn the machine on and insert the tape cartridge in the T-15 tape drive.

(2) Type GET "VCSCH1", then press EXECUTE.

(3) When the program is loaded in the machine memory (green square light at the bottom right-hand corner of the screen goes out), press RUN.

The following message will appear on the screen:

THIS PROGRAM COMPUTES AND PLOTS FULL SCHLUMBERGER SOUNDING CURVES EXPANDED AT RIGHT ANGLES TO A VERTICAL DIKE, and at the bottom of the screen will appear the first prompting statement: DO YOU WISH TO HAVE A CURVE SET ? (Y/N) FOR ONE CURVE ONLY TYPE N FOR NO.

Type Y (for YES, meaning you wish to compute a master set) or type N (for NO, meaning you wish to compute only a particular single sounding curve). Then press CONT.

(4) Subsequent prompting statements assisted by illustrations on the screen are self explanatory, and also are shown in the examples section.

(5) To obtain a printed copy of the numerical values for the computed sounding curve wait until the prompting statement INPUT MIN AB/2, MAX AB/2 appears on the screen. Then:

(i) type PRINTER IS 0 and press EXECUTE,

(ii) type the numerical values for the smallest and largest AB/2 values (separated by commas) and press CONT.

To exit from the printer mode, type PRINTER IS 16 and press EXECUTE.

- (6) To obtain a hard copy of a single curve, type DUMP GRAPHICS and press EXECUTE.

The graphic display is not cleared after plotting or dumping a single sounding curve. This allows the user to plot other curves with different model parameters on top of the previous plots. To do this press CONT and enter the new parameters for the new model. To clear the present plot, type GCLEAR and press EXECUTE. Press CONT to start a new problem.

- (7) Upon the completion and plotting of a complete set of master curves, four beeps will be heard and a full-screen cross will appear on the screen. The location of the cross on the plot can be moved using the up, down, right, and left arrows on the DISPLAY buttons. The cross is used to indicate to the computer the appropriate position for labeling the different curves with the proper values of  $d_2$ . This is done as follows: (i) move the cross to the appropriate location on the screen and press CONT. The prompting statement: TYPE VALUE FOR  $D_2$ , OR TYPE N (WHEN DONE LABELING) will appear. (ii) Type the appropriate value for  $d_2$  (or type appropriate symbols such as vertical or slanted lines to be used for leaders to the curves) and press CONT. (iii) Repeat step (i) as desired to label all the other curves. Note that PEN-1 and PEN 1 commands can be used to correct errors in labeling. (iv) Upon completion of the labeling operation, type N then press CONT. The prompting statement: DUMP GRAPHICS ? (Y/N) will appear on the screen. (v) For a hard copy of the labeled set of master curves type Y and press CONT.
- If the user types N and presses CONT, the program will start from the beginning, and the graphics will be deleted.
- (8) To exit from the program press STOP.

# EXAMPLE OF THE COMPUTATION OF A SINGLE SOUNDING CURVE

Follow steps 1, 2, and 3 in the section on instructions for users.

STEP	PROMPTING STATEMENT	TYPE THEN PRESS CONT
(1)	DO YOU WISH TO HAVE A CURVE SET ? (Y/N)	N
(2)	DESIGNATE MEDIUM WHERE FIXED CENTER IS LOCATED: 1, 2, or 3	1
(3)	INPUT VALUES FOR RHO-1,RHO-2,RHO-3	5,200,25
(4)	INPUT DISTANCES TO CONTACTS:D1,D2	30,60
(5)	INPUT MIN (AB/2), MAX (AB/2)	1,10000
(6)	0:0 COORDINATES ? PRESS CONT FOR AUTO SCALE	(PRESS CONT)

## NOTE:

- (1) To obtain a printed listing of the computed apparent resistivities for the above example, at step (5) first type PRINTER IS 0, press EXECUTE, then type the minimum and maximum AB/2 values and press CONT.  
When the curve is plotted type PRINTER IS 16, press EXECUTE, to exit from printer mode.
- (2) In step (6) above, the 0:0 COORDINATES are the values of the smallest AB/2 and smallest apparent resistivity that the user may wish to assign to the first abscissa and ordinate to shift the plotted curve upward, downward, to the right or to the left. Otherwise by pressing CONT the program will attempt to find the proper coordinate values that will result in a properly located curve.

- (3) To obtain a hard copy, type DUMP GRAPHICS and press EXECUTE.
- (4) Type GCLEAR and press EXECUTE to clear the screen for the next problem if so desired.
- (5) The printer output and the plotted curve for the above example are given in table 1 and in figure 1, respectively.



Table 1 -- Printer output for a single sounding curve.

RHO-1;RHO-2;RHO-3	5.0000	200.0000	25.0000
CENTER IS ON	1.0000		
DISTANCES=	30.0000	60.0000	

AB/2	APPARENT RESISTIVITY
1.0000	5.0000
1.1220	5.0000
1.2589	5.0000
1.4125	5.0000
1.5849	5.0000
1.7783	5.0000
1.9953	5.0001
2.2387	5.0002
2.5119	5.0004
2.8184	5.0008
3.1623	5.0015
3.5481	5.0026
3.9811	5.0043
4.4668	5.0066
5.0119	5.0096
5.6234	5.0136
6.3096	5.0184
7.0795	5.0242
7.9433	5.0311
8.9125	5.0395
10.0000	5.0502
11.2202	5.0653
12.5893	5.0881
14.1254	5.1244
15.8489	5.1824
17.7828	5.2743
19.9526	5.4180
22.3872	5.6432
25.1189	6.0028
28.1838	6.5989
30.0000	7.0882
33.6606	6.9416
37.7678	6.7149
42.3761	6.3628
47.5468	5.8081
53.3484	4.9121
59.8579	3.4083
60.0000	3.3678
67.3211	3.3494
75.5355	3.3324

84.7523	3.3174
95.0936	3.3052
106.6968	3.2966
119.7157	3.2923
134.3233	3.2927
150.7132	3.2985
169.1030	3.3100
189.7367	3.3275
212.8880	3.3511
238.8643	3.3807
268.0102	3.4162
300.7123	3.4572
337.4048	3.5034
378.5744	3.5542
424.7675	3.6090
476.5969	3.6672
534.7506	3.7281
600.0000	3.7910
673.2111	3.8552
755.3552	3.9201
847.5225	3.9850
950.9359	4.0494
1066.9676	4.1127
1197.1574	4.1745
1343.2327	4.2343
1507.1319	4.2920
1691.0298	4.3471
1897.3666	4.3996
2128.8803	4.4493
2388.6430	4.4960
2680.1016	4.5398
3007.1234	4.5807
3374.0479	4.6186
3785.7441	4.6538
4247.6747	4.6863
4765.9694	4.7161
5347.5056	4.7435
6000.0000	4.7685
6732.1107	4.7914
7553.5525	4.8121
8475.2253	4.8310
9509.3591	4.8482




Figure 1: Output example for a single sounding curve.

# EXAMPLE OF THE COMPUTATION OF A SET OF MASTER CURVES

Follow steps 1, 2, and 3 given in section on instruction for users.

-----		
STEP	PROMPTING STATEMENT	TYPE THEN PRESS CONT
-----		
(1)	DO YOU WISH TO HAVE A CURVE SET ? (Y/N)	Y
(2)	DESIGNATE MEDIUM WHERE FIXED CENTER IS LOCATED: 1, 2, OR 3	2
(3)	INPUT VALUES FOR RHO-1,RHO-2,RHO-3	0.1,1,0.1
-----		

Figure 2 shows the output to the above example after using

the labeling technique described earlier.




Figure 2: Output example for a set of master curves.

#### ACKNOWLEDGMENT

The author wishes to thank his colleague R. J. Bisdorf for his guidance and assistance in programming the HP-9845 desk-top computer at the early stages of this effort.

## REFERENCES

- Abramowitz, Milton, and Stegun, I.A. (Editors), 1964, Handbook of mathematical functions with formulas, graphs, and mathematical tables: National Bureau of Standards, Applied Mathematics Series 55, 1045 p.
- Hedstrom, Helmer, 1932, Electrical prospecting for auriferous quartz veins and reefs: Mining Magazine, v. 46, p. 201-213.
- Kunetz, Geza, 1955, Einfluss vertikaler Schichten auf elektrische Sondierungen: Zeitschrift fur Geophysik, v. 21, p. 10-24.
- Logn, O., 1954, Mapping nearly vertical discontinuities by earth resistivities: Geophysics, v. 19, p. 739-760.
- Stefanescu, Sabba, Schlumberger, Conrad, and Schlumberger, Marcel, 1930, Sur la distribution electrique potentielle autour d'une prise de terre ponctuelle dans un terrain a couches horizontales, homogenes et isotropes: Jour. Physique et Radium, v. 1, p.132-140.
- Van Nostrand, R.G., and Cook, K.L., 1966, Interpretation of resistivity data: U.S. Geol. Survey Prof. Paper 499, 310 p.
- Zohdy, A.A.R., 1970, Variable azimuth Schlumberger resistivity sounding and profiling near a vertical contact: U.S. Geol. Survey Bull. 1313-A, 22 p.

APPENDIX I  
MATHEMATICAL FORMULAS



## MATHEMATICAL FORMULAS

The Gauss-Laguerre numerical integration formula (Abramowitz and Stegun, 1964, p.923) states that:

$$\int_0^{\infty} e^{-x} f(x) dx \approx \sum_{i=1}^n W_i f(x_i) ,$$

where  $W_i$  = Laguerre weights,  
 $x_i$  = Laguerre abscissas,  
 $f(x_i)$  = function to be evaluated at proper  $x_i$  values.

By comparing results obtained from the above formula with those obtained from a closed algebraic expression for the limiting case of a vertical fault (Zohdy, 1970), and by experimenting with various values of  $n$ , I found that the computations are sufficiently accurate with  $n=5$ . In fact with values as low as  $n=1$  and  $W_i=1$  the shape of the curve for many models is almost acceptable for practical purposes with errors of about 30 percent in parts of it. The Laguerre abscissas and weights are given in steps 150 to 250 of the program (see appendix II) and in the above reference.

Two sets of expressions must be derived, one for the sounding center on medium 1 (left layer), and the other for the sounding center on medium 2 (middle layer). The expressions for the sounding center on medium 3 (right layer) are identical to those with the center on medium 1, but with the resistivities of medium 1 and medium 3 exchanged.

# I. SOUNDING CENTER ON MEDIUM 1:

In the following equations, the following definitions apply (also see figure 3):

$\rho_1, \rho_2, \rho_3$  = resistivities of medium 1, 2, and 3, respectively

$$k_{21} = (\rho_2 - \rho_1) / (\rho_2 + \rho_1)$$

$$k_{32} = (\rho_3 - \rho_2) / (\rho_3 + \rho_2)$$

$d_1, d_2$  = distances from sounding center to first and

$y$  =  $AB/2$  = Schlumberger electrode spacing; that is half distance between current electrodes A and B.

$\bar{\rho}_s(y)$  = Schlumberger apparent resistivity as a function of  $y=AB/2$ .

(i) All electrodes on medium 1 ( $d_2 > y > d_1$ ):

$$\bar{\rho}_s(y) = \frac{\rho_1}{2} \sum_i^5 W_i f(x_i),$$

where

$$f(x_i) = x_i \left[ 2 + \frac{[k_{21} + k_{32} e^{-\frac{2x_i}{y}(d_2-d_1)}] [e^{-\frac{2x_i}{y}(d_1-y)} - e^{-\frac{2x_i}{y}d_1}]}{1 + k_{21} k_{32} e^{-\frac{2x_i}{y}(d_2-d_1)}} \right].$$

(ii) Electrode B on medium 2, all others on medium 1 ( $d_2 > y > d_1$ ):

$$\bar{\rho}_s(y) = \frac{\rho_1}{2} \sum_i^5 W_i f(x_i),$$

where


$$f(x_i) = x_i \left[ \frac{\rho_2}{\rho_1} \left[ 1 + \frac{k_{32} [e^{-\frac{2x_i}{y}(d_2-y)} - k_{21} e^{-\frac{2x_i}{y}(d_2-d_1)}] - k_{21} [1 + k_{32} e^{-\frac{2x_i}{y}(d_2-y)}]}{1 + k_{21} k_{32} e^{-\frac{2x_i}{y}(d_2-d_1)}} \right] \right. \\ \left. + \left[ 1 - \frac{e^{-\frac{2x_i}{y}d_1} [k_{21} + k_{32} e^{-\frac{2x_i}{y}(d_2-d_1)}]}{1 + k_{21} k_{32} e^{-\frac{2x_i}{y}(d_2-d_1)}} \right] \right].$$

(iii) Electrode B on medium 3, all others on medium 1 ( $d_2 < y < d_1$ ):

$$\bar{P}_3(y) = \frac{\rho_1}{2} \sum_i^5 w_i f(x_i)$$

where

$$f(x_i) = x_i \left[ \left[ 1 - \frac{e^{-\frac{2x_i}{y}d_1} [k_{21} + k_{32} e^{-\frac{2x_i}{y}(d_2-d_1)}]}{1 + k_{21} k_{32} e^{-\frac{2x_i}{y}(d_2-d_1)}} \right] + \frac{\rho_3}{\rho_1} \left[ 1 - \frac{k_{32} + k_{21}[1 - k_{32}] + k_{21} k_{32} e^{-\frac{2x_i}{y}(d_2-d_1)}}{1 + k_{21} k_{32} e^{-\frac{2x_i}{y}(d_2-d_1)}} \right] \right]$$



Figuro 3: Electrode array center on medium 1.

## II. SOUNDING CENTER ON MEDIUM 2:

In the following equations, the same definitions for the variables apply, except for  $d_1$  and  $d_2$  (see figure 4) where:

$d_1$  = distance from sounding center to the LEFT contact separating RH0-1 from RH0-2,

$d_2$  = distance from sounding center to the RIGHT contact separating RH0-2 from RH0-3.

(i) All electrodes on medium 2 ( $d_2 > y < d_1$ )

$$\bar{P}_S(y) = \frac{\rho_2}{2} \sum_i^S W_i f(x_i),$$

where

$$f(x_i) = x_i \left[ 2 + \frac{k_{21} \left[ e^{-\frac{2x_i d_1}{y}} + k_{32} e^{-\frac{2x_i (d_1 + d_2 - y)}{y}} \right] + k_{32} \left[ e^{-\frac{2x_i (d_2 - y)}{y}} - k_{21} e^{-\frac{2x_i (d_2 + d_1)}{y}} \right]}{1 + k_{21} k_{32} e^{-\frac{2x_i (d_2 + d_1)}{y}}} \right. \\ \left. - \frac{k_{21} \left[ e^{-\frac{2x_i (d_1 - y)}{y}} + k_{32} e^{-\frac{2x_i (d_2 + d_1)}{y}} \right] + k_{32} \left[ e^{-\frac{2x_i d_2}{y}} - k_{21} e^{-\frac{2x_i (d_1 + d_2 - y)}{y}} \right]}{1 + k_{21} k_{32} e^{-\frac{2x_i (d_2 + d_1)}{y}}} \right]$$

(ii) Electrode A on medium 1, all others on medium 2 ( $d_1 < y < d_2$ )

$$\bar{P}_S(y) = \frac{\rho_2}{2} \sum_i^S W_i f(x_i),$$

where

$$f(x_i) = x_i \left[ \frac{\rho_1}{\rho_2} \left[ 1 + \frac{k_{21} \left[ 1 - k_{32} e^{-\frac{2x_i (d_1 + d_2)}{y}} \right] - k_{32} \left[ 1 + k_{21} \right] e^{-\frac{2x_i d_2}{y}}}{1 + k_{21} k_{32} e^{-\frac{2x_i (d_2 + d_1)}{y}}} \right] \right. \\ \left. + \left[ 1 + \frac{k_{21} \left[ e^{-\frac{2x_i d_1}{y}} + k_{32} e^{-\frac{2x_i (d_1 + d_2 - y)}{y}} \right] + k_{32} \left[ e^{-\frac{2x_i (d_2 - y)}{y}} - k_{21} e^{-\frac{2x_i (d_1 + d_2)}{y}} \right]}{1 + k_{21} k_{32} e^{-\frac{2x_i (d_2 + d_1)}{y}}} \right] \right]$$

(iii) Electrode B on medium 3, all others on medium 2 ( $d_2 < y < d_1$ ):

$$\bar{P}_3(y) = \frac{\rho_2}{2} \sum_1^5 W_i f(x_i),$$

where

$$f(x_i) = x_i \left[ 1 - \frac{k_{21} \left[ e^{-\frac{2x_i(d_1-y)}{y}} + k_{32} e^{-\frac{2x_i(d_1+d_2)}{y}} \right] + k_{32} \left[ e^{-\frac{2x_i d_2}{y}} - k_{21} e^{-\frac{2x_i(d_1+d_2-y)}{y}} \right]}{1 + k_{21} k_{32} e^{-\frac{2x_i}{y}(d_2+d_1)}} \right. \\ \left. + \frac{\rho_3}{\rho_2} \left[ 1 + \frac{k_{21} [1 - k_{32}] e^{-\frac{2x_i d_1}{y}} - k_{32} [1 + k_{21}] e^{-\frac{2x_i}{y}(d_2+d_1)}}{1 + k_{21} k_{32} e^{-\frac{2x_i}{y}(d_1+d_2)}} \right] \right]$$

(iv) Electrode A on medium 1, electrode B on 3, and center on 2 ( $d_2 < y < d_1$ ):

$$\bar{P}_3(y) = \frac{\rho_2}{2} \sum_1^5 W_i f(x_i),$$

where

$$f(x_i) = x_i \left[ \frac{\rho_1}{\rho_2} \left[ 1 + \frac{k_{21} [1 - k_{32}] e^{-\frac{2x_i}{y}(d_1+d_2)} - k_{32} [1 + k_{21}] e^{-\frac{2x_i d_2}{y}}}{1 + k_{21} k_{32} e^{-\frac{2x_i}{y}(d_1+d_2)}} \right] \right. \\ \left. + \frac{\rho_3}{\rho_2} \left[ 1 + \frac{k_{21} [1 - k_{32}] e^{-\frac{2x_i d_1}{y}} - k_{32} [1 + k_{21}] e^{-\frac{2x_i}{y}(d_1+d_2)}}{1 + k_{21} k_{32} e^{-\frac{2x_i}{y}(d_1+d_2)}} \right] \right]$$




Figure 4: Electrode array center on medium 2.

APPENDIX II  
PROGRAM LISTING OF VCSCH1

```

10  ! ****PROGRAM FOR COMPUTING AND PLOTTING FULL SCHLUMBERGER****
20  ! ****SOUNDING CURVES EXPANDED AT RIGHT ANGLES TO A VERTICAL****
30  ! ****DIKE. BY ADEL R.P. CONDY, USGS, DENVER, COLORADO, SEP.13,79****
40  ! ****
50  ! ****
60  !
70  !
80  !
90  !
100 OVERLAP
110 OPTION BASE 1
120 FIXED 4
130 DIM Ki(5),Wt(5),Fi(5),RhoBar(100),Rad(100)
140 MAT READ Ki          ! LA GUERRE ABSCISSAS FOR N=5
150   DATA 0.263560319718
160   DATA 1.413403059107
170   DATA 3.596425771041
180   DATA 7.085818005859
190   DATA 12.640800844276
200 MAT READ Wt          ! LA GUERRE WEIGHTS FOR N=5
210   DATA 5.21755610583E-01
220   DATA 3.98666811083E-01
230   DATA 7.59424496817E-02
240   DATA 3.61175867992E-03
250   DATA 2.33699723858E-05
260   EXIT GRAPHICS
270   PRINT PAGE
280   PRINT "THIS PROGRAM COMPUTES AND PLOTS FULL SCHLUMBERGER SOUNDING CURVES
EXPANDED AT RIGHT ANGLES TO A VERTICAL DIKE"
290   PRINT
300   INPUT "DO YOU WISH TO HAVE A CURVE SET ? (Y/N). FOR ONE CURVE ONLY TYPE N
FOR NO.",Curve_set$
310   IF (Curve_set$="Y") OR (Curve_set$="N") THEN 360
320   DISP "PLEASE TYPE: Y OR N .. (FOR YES OR NO)"
330   BEEP
340   WAIT 2000
350   GOTO 300
360   PRINT PAGE
370   CALL Plot
380   INPUT "DESIGNATE MEDIUM WHERE FIXED CENTER IS LOCATED: 1,2,OR 3",Center
390   IF (Center=1) OR (Center=2) OR (Center=3) THEN 440
400   DISP "ERROR: CENTER MUST BE ON MEDIUM 1,2,OR 3"
410   BEEP
420   WAIT 3000
430   GOTO 380
440   IF Center=1 THEN CALL Plot1
450   IF Center=2 THEN CALL Plot2
460   IF Center=3 THEN CALL Plot3
470   INPUT "INPUT VALUES FOR RHO-1,RHO-2,RHO-3",R1,R2,R3
480   IF (R1=0) AND (Center=1) THEN 520
490   IF (R2=0) AND (Center=2) THEN 520
500   IF (R3=0) AND (Center=3) THEN 520
510   GOTO 590
520   DISP "TRIVIAL CASE, WITH CENTER ON ZERO RESISTIVITY ...."
530   BEEP
540   WAIT 3000
550   DISP "ALL APPARENT RESISTIVITIES WILL BE ZERO !!!"
560   WAIT 3000
570   BEEP
580   GOTO 470
590   IF Curve_set$="N" THEN 660
600   D1=1
610   READ D2
620   DATA 40,20,8,4,2,1.5,1,1000000
630   Ymir=.1

```

```

640 Ymax=1000
650 GOTO 1020
660 IF (Center=1) OR (Center=3) THEN 750
670 INPUT "INPUT DISTANCES TO THE LEFT, AND TO THE RIGHT: D1,D2",D1,D2
680 IF (D1<>0) AND (D2<>0) THEN 860
690 DISP "ERROR: D1 OR D2 CANNOT BE ZERO"
700 BEEP
710 WAIT 2000
720 DISP "BECAUSE OF DISCONTINUITY IN APPARENT RESISTIVITY"
730 WAIT 3000
740 GOTO 670
750 INPUT "INPUT DISTANCES TO CONTACTS: D1,D2 ",D1,D2
760 IF D1<>0 THEN 810
770 DISP "ERROR: D1 CANNOT BE ZERO BECAUSE OF DISCONTINUITY IN APPARENT RESI
STIVITY"
780 BEEP
790 WAIT 3000
800 GOTO 750
810 IF D2>D1 THEN 860
820 DISP " ERROR: D2 MUST BE LARGER THAN OR EQUAL TO D1"
830 BEEP
840 WAIT 3000
850 GOTO 750
860 INPUT "INPUT MINIMUM AB/2,MAXIMUM AB/2",Ymin,Ymax
870 IF Ymax>Ymin THEN 920
880 DISP "ERROR: MIN AB/2 MUST BE SMALLER THAN MAX AB/2"
890 BEEP
900 WAIT 3000
910 GOTO 860
920 PRINT "RHO-1;RHO-2;RHO-3",R1;R2;R3
930 PRINT "CENTER IS ON",Center
940 IF (Center=1) OR (Center=3) THEN 980
950 PRINT "DISTANCE TO THE LEFT=";D1
960 PRINT "DISTANCE TO THE RIGHT=";D2
970 GOTO 990
980 PRINT "DISTANCES=",D1;D2
990 PRINT
1000 PRINT " AB/2","APPARENT RESISTIVITY"
1010 PRINT
1020 Y=Ymin ! SMALLEST VALUE FOR AB/2
1030 Ncy=3 ! #OF LOG CYCLES ON Y-AXIS
1040 Ncx=4 ! #OF LOG CYCLES ON X-AXIS
1050 P=20 ! #OF POINTS PER LOG CYCLE
1060 K=0
1070 Kk=0
1080 R=EXP(LOG(10)/P) !ABSCISSA RATIO FOR SUCCESSIVE POINTS
1090 J=1
1100 FOR I=1 TO 5
1110 Fi(I)=-2*Ki(I)
1120 NEXT I
1130 IF (R1=0) AND (R2=0) THEN GOTO 1160
1140 K21=(R2-R1)/(R2+R1)
1150 GOTO 1170
1160 K21=0
1170 IF (R2=0) AND (R3=0) THEN 1200
1180 K32=(R3-R2)/(R3+R2)
1190 GOTO 1210
1200 K32=0
1210 IF Center=1 THEN C1
1220 IF Center=2 THEN C2
1230 IF Center=3 THEN C3
1240 C2: Total=0
1250 FOR I=1 TO 5
1260 E1=EXP(Fi(I)*D1/Y)
1270 E2=EXP(Fi(I)*D2/Y)

```



```

1280 E3=EXP(Fi(I)*(D1+D2)/Y)
1290 E4=EXP(Fi(I)*(D1-Y)/Y)
1300 E5=EXP(Fi(I)*(D2-Y)/Y)
1310 E6=EXP(Fi(I)*(D1+D2-Y)/Y)
1320 D=1+K21*K32*E3
1330 IF Y>D2 THEN D
1340 A: A=(K21*(E1+K32*E6)+K32*(E5-K21*E3))/D
1350 A=1+A
1360 IF Y>D1 THEN C
1370 B: B=(K21*(E4+K32*E3)+K32*(E2-K21*E6))/D
1380 B=1+B
1390 IF Y>D2 THEN L3
1400 GOTO L1
1410 C: C=(K21*(1-K32*E3)-K32*(1+K21)*E2)/D
1420 C=1+C
1430 C=R1*C/R2
1440 IF Y>D2 THEN L4
1450 GOTO L2
1460 D: F=(K21*(1-K32)*E1-K32*(1+K21*E3))/D
1470 F=1+F
1480 F=R3*F/R2
1490 IF Y>D1 THEN C
1500 GOTO B
1510 L1: Total=Total+Wt(I)*Ki(I)*(A+B)
1520 GOTO L5
1530 L2: Total=Total+Wt(I)*Ki(I)*(A+C)
1540 GOTO L5
1550 L3: Total=Total+Wt(I)*Ki(I)*(B+F)
1560 GOTO L5
1570 L4: Total=Total+Wt(I)*Ki(I)*(C+F)
1580 L5: NEXT I
1590 Rad(J)=Y
1600 Rhobar(J)=R2*Total/2
1610 PRINT Rad(J),Rhobar(J)
1620 J=J+1
1630 IF Y>Ymax THEN GOTO G1
1640 Y=Y*R
1650 IF Y<D1 THEN 1700
1660 K=K+1
1670 IF K>1 THEN 1700
1680 Y=D1
1690 GOTO C2
1700 IF Y<D2 THEN C2
1710 Kk=Kk+1
1720 IF Kk>1 THEN C2
1730 Y=D2
1740 GOTO C2
1750 C1: IF Y>D1 THEN C12
1760 Total=0
1770 FOR I=1 TO 5
1780 E1=EXP(Fi(I)*(D2-D1)/Y)
1790 E2=EXP(Fi(I)*(D1-Y)/Y)
1800 E3=EXP(Fi(I)*D1/Y)
1810 D=1+K21*K32*E1
1820 A=2+(K21+K32*E1)*(E2-E3)/D
1830 Total=Total+Wt(I)*Ki(I)*A
1840 NEXT I
1850 Rhobar(J)=R1*Total/2
1860 Rad(J)=Y
1870 PRINT Rad(J),Rhobar(J)
1880 J=J+1
1890 Y=Y*R
1900 IF Y>Ymax THEN Reset
1910 GOTO C1
1920 C12: K=K+1

```

```

1930 IF K=1 THEN Y=D1
1940 IF Y>D2 THEN C13
1950 Total=0
1960 FOR I=1 TO 5
1970 E1=EXP(F1(I)*(D2-D1)/Y)
1980 E2=EXP(F1(I)*D1/Y)
1990 D=1+K21+K32+E1
2000 A=(K21+K32+E1)/D
2010 A=A+E2
2020 A=1-A
2030 E3=EXP(F1(I)*(D2-D1)/Y)
2040 E4=EXP(F1(I)*(D2-Y)/Y)
2050 B=(K21+K32*(K21*(E3+E4)-E4))/D
2060 B=R2*(1-B)/R1
2070 Total=Total+Wt(I)*Ki(I)*(A+B)
2080 NEXT I
2090 Rad(J)=Y
2100 Rhobar(J)=R1*Total/2
2110 PRINT Rad(J),Rhobar(J)
2120 J=J+1
2130 Y=Y*R
2140 IF Y>Ymax THEN Reset
2150 GOTO C12
2160 C13: Kk=Kk+1
2170 IF Kk=1 THEN Y=D2
2180 Total=0
2190 FOR I=1 TO 5
2200 E1=EXP(F1(I)*(D2-D1)/Y)
2210 E2=EXP(F1(I)*D1/Y)
2220 D=1+K21+K32+E1
2230 A=1-E2*(K21+K32+E1)/D
2240 B=R3*(1-(K32+K21*(1-K32)+K32*K21+E1)/D)/R1
2250 Total=Total+Wt(I)*Ki(I)*(A+B)
2260 NEXT I
2270 Rhobar(J)=R1*Total/2
2280 Rad(J)=Y
2290 PRINT Rad(J),Rhobar(J)
2300 J=J+1
2310 Y=Y*R
2320 IF Y>Ymax THEN Reset
2330 GOTO C13
2340 C3: H=R1 ! EXCHANGE VALUES OF R1 AND R3
2350 R1=R3 ! AND MAKE Center =1 INSTEAD OF 3
2360 R3=H
2370 Center=1
2380 Z=1
2390 GOTO 1140
2400 Reset: ! RE-SET VALUES OF R1,R3,AND Center
2410 IF Z<>1 THEN GOTO G1
2420 H=R1
2430 R1=R3
2440 R3=H
2450 Center=3
2460 Z=0
2470 G1: CALL G1(Ncx,Ncy,Rad(*),Rhobar(*),J,R1,R2,R3,Center,D1,D2,Curve_set#)
2480 IF Curve_set#="N" THEN 2720
2490 Curve_set#="Y+1"
2500 IF D2=1000000 THEN 2520
2510 GOTO 610
2520 BEEP
2530 WAIT 200
2540 BEEP
2550 WAIT 200
2560 BEEP
2570 DIGITICE X,Y

```

```

2580 INPUT "TYPE VALUE FOR D2, OR TYPE N (WHEN DONE LABELING)",D2$
2590 IF D2#="N" THEN 2640
2600 LOG 5
2610 MOVE X,Y
2620 LABEL D2$
2630 GOTO 2570
2640 INPUT "DUMP GRAPHICS ? (Y/N)",Dump_graphics$
2650 IF Dump_graphics#="N" THEN 2670
2660 DUMP GRAPHICS
2670 GCLEAR
2680 PRINT PAGE
2690 RESTORE 620
2700 EXIT GRAPHICS
2710 GOTO 300
2720 END

```

```

2730 SUB Plot2
2740 PRINT PAGE
2750 PRINT LIN(3)
2760 PRINT "
2770 PRINT "
2780 PRINT "
2790 PRINT "
2800 PRINT "
2810 PRINT "
2820 PRINT "
2830 PRINT "
2840 PRINT "
2850 PRINT "
2860 PRINT "
2870 SUBEND

```

FIXED CENTER		
A	MN	B
<-- <--!	!!	! --> -->
	<-D1-><-----D2----->	
	////////////////////	
	////////////////////	
RHO-1	//// RHO-2 ////	RHO-3
	////////////////////	
	////////////////////	
	////////////////////	

```

2880 SUB Plot
2890 PRINT
2900 PRINT
2910 PRINT "
2920 PRINT "
2930 PRINT "
2940 PRINT "
2950 PRINT "
2960 PRINT "
2970 PRINT "
2980 PRINT "
2990 SUBEND

```

1	2	3

```

3000 SUB Plot1
3010 PRINT PAGE
3020 PRINT LIN(3)
3030 PRINT "
3040 PRINT "
3050 PRINT "
3060 PRINT "
3070 PRINT "
3080 PRINT "
3090 PRINT "
3100 PRINT "
3110 PRINT "
3120 PRINT "
3130 PRINT "
3140 PRINT "
3150 PRINT "
3160 SUBEND

```

FIXED CENTER		
A	MN	B
<- <- !	!!	! -> ->
	<---D1--->	////////////////////
	<---D2--->	----->
		////////////////////
		////////////////////
		////////////////////
RHO-1	//// RHO-2 ////	RHO-3
	////////////////////	
	////////////////////	
	////////////////////	

```

3170 SUB Plot3
3180 PRINT PAGE
3190 PRINT LIN(3)
3200 PRINT "
3210 PRINT "
3220 PRINT "

```

FIXED CENTER		
A	MN	B
!	!!	!

```

3230 PRINT "
3240 PRINT "
3250 PRINT "
3260 PRINT "
3270 PRINT "
3280 PRINT "
3290 PRINT "   RHO-1   RHO-2   RHO-3
3300 PRINT "
3310 PRINT "
3320 PRINT "
3330 SUBEND
3340 SUB G1(Ncx,Ncy,Rad(*),Rhobar(*),J,R1,R2,R3,Center,D1,D2,Curve_set$) ! GRAP
HICS
3350 ! Ncx=NUMBER OF LOG CYCLES ON X-AXIS=4
3360 ! Ncy=NUMBER OF LOG CYCLES ON Y-AXIS=3
3370 ! Rad=ELECTRODE SPACINGS (AB/2)
3380 ! Rhobar=APPARENT RESISTIVITIES
3390 ! J=NUMBER OF POINTS ON THE CURVE
3400 ! R1,R2,R3= TRUE RESISTIVITIES
3410 ! Center=1,2,OR 3=LOCATION OF FIXED CENTER ON MEDIUM 1,2,OR 3
3420 ! D1,D2= DISTANCES TO CONTACTS
3430 DIM Ax(5),Ay(4)
3440 GRAPHICS
3450 IF Curve_set$="Y+1" THEN 4580
3460 Ax(1)=0
3470 Ay(1)=0
3480 IF Curve_set$="Y" THEN 3540
3490 INPUT "0:0 COORDINATES ? PRESS CONT FOR AUTO SCALE",Ax(1),Ay(1)
3500 IF Ax(1)=0 THEN 3540
3510 Ax(1)=10^INT(LGT(Ax(1)))
3520 GCLEAR
3530 GOTO 3550
3540 Ax(1)=10^INT(LGT(Rad(1)))
3550 FOR I=2 TO 5
3560 Ax(I)=Ax(I-1)*10
3570 NEXT I
3580 Minrho=Rhobar(1)
3590 Maxrho=Rhobar(1)
3600 FOR I=1 TO J-1 !COMPUTE MINIMUM AND MAXIMUM Rhobar.
3610 IF Rhobar(I)<Minrho THEN Minrho=Rhobar(I)
3620 IF Rhobar(I)>Maxrho THEN Maxrho=Rhobar(I)
3630 NEXT I
3640 Average_rho=EXP((LOG(Rhobar(1))+LOG(Minrho)+LOG(Maxrho))/3)
3650 GOTO 3680
3660 Ay(1)=10^INT(LGT(Ay(1)))
3670 GOTO 3690
3680 Ay(1)=10^INT(LGT(Average_rho))/10
3690 FOR I=2 TO 4
3700 Ay(I)=Ay(I-1)*10
3710 NEXT I
3720 R1=10^(LGT(Ax(1))+Ncx)
3730 R2=10^(LGT(Ay(1))+Ncy)
3740 GRAPHICS
3750 LOCATE 7,115,5,100
3760 CALL Log_cycle(LGT(Ax(1)),LGT(R1),LGT(Ay(1)),LGT(R2),Ncx,Ncy)
3770 SETGO
3780 LOGO 1
3790 X=90 !DRAW DIKE SKETCH
3800 Y=85
3810 MOVE X,Y
3820 DRAW X+20,Y
3830 MOVE X+7,Y
3840 DRAW X+7,Y-7
3850 MOVE X+13,Y
3860 DRAW X+13,Y-7

```

```

3870 IF Center=2 THEN 3900
3880 IF Center=3 THEN 4030
3890 MOVE X+2.7,Y
3900 DRAW X+3.7,Y+3
3910 DRAW X+1.7,Y+3
3920 DRAW X+2.7,Y
3930 MOVE X-1.8,Y+1
3940 LABEL "<- ->"
3950 GOTO Label11
3960 MOVE X+10,Y
3970 DRAW X+11,Y+3
3980 DRAW X+9,Y+3
3990 DRAW X+10,Y
4000 MOVE X+5.3,Y+1
4010 LABEL "<- ->"
4020 GOTO Label11
4030 MOVE X+15,Y
4040 DRAW X+16,Y+3
4050 DRAW X+14,Y+3
4060 DRAW X+15,Y
4070 MOVE X+10.5,Y+1
4080 LABEL "<- ->"
4090 Label11: !
4100 X=20
4110 Y=95
4120 FIXED 2
4130 MOVE X,Y
4140 IF Curve_set#="N" THEN 4170
4150 LABEL "D1=1; D2 = as labeled" RHO: "&VAL$(R1)&","&VAL$(R2)&","&VAL$(R3)
)
4160 GOTO 4180
4170 LABEL "D1,D2: "&VAL$(D1)&","&VAL$(D2)&" RHO: "&VAL$(R1)&","&VAL$(R2)&","
&VAL$(R3)
4180 X=7
4190 Y=6
4200 MOVE X,Y
4210 FOR I=1 TO 5
4220 N=0
4230 K=1
4240 IF Ax(I)>=K THEN Label12
4250 K=K/10
4260 N=N+1
4270 GOTO 4240
4280 Label12: FIXED N
4290 LONG 4
4300 LABEL VAL$(Ax(I))
4310 X=X+27
4320 MOVE X,Y
4330 NEXT I
4340 MOVE 25,2
4350 LONG 1
4360 LABEL " SCHLUMBERGER ELECTRODE SPACING (AB/2)"
4370 MOVE 0,33
4380 DEG
4390 LONG 3
4400 LDIR 90
4410 LABEL "APPARENT RESISTIVITY"
4420 X=3
4430 Y=12
4440 MOVE X,Y
4450 FOR I=1 TO 4
4460 K=1
4470 N=0
4480 IF Ay(I)>=K THEN Label13
4490 K=K/10

```

```

4500 N=N+1
4510 GOTO 4480
4520 Labels: FIXED N
4530 LORG 6
4540 LABEL VAL#(Ay(I))
4550 Y=Y+27
4560 MOVE X,Y
4570 NEXT I
4580 LDIR 0
4590 LORG 1
4600 FIXED 4
4610 SETUU
4620 MOVE LGT(Rad(1)),LGT(Rhobar(1))
4630 FOR I=1 TO J-1
4640 PLOT LGT(Rad(I)),LGT(Rhobar(I))
4650 NEXT I
4660 BEEP
4670 WAIT 100
4680 BEEP
4690 SUBEND
4700 SUB Log_cycle(Lxmin,Lxmax,Lymin,Lymax,Ncx,Ncy)
4710 LINE TYPE 7
4720 SHOW Lxmin,Lxmax,Lymin,Lymax
4730 MOVE Lxmin,Lymin
4740 FOR I=1 TO Ncx+1
4750 X=Lxmin+I-1
4760 PLOT X,Lymin,1
4770 PLOT X,Lymax,2
4780 NEXT I
4790 FOR I=1 TO Ncy+1
4800 Y=Lymin+I-1
4810 PLOT Lxmin,Y,1
4820 PLOT Lxmax,Y,2
4830 NEXT I
4840 LINE TYPE 1
4850 SUBEND

```